

Tip des Monats: Player-Vertikal- bewegung

Bei den ATARI-Heimcomputern ist kein Register zur Vertikalbewegung der Player vorgesehen, wie zum Beispiel die Adressen 53248–53250 für die horizontale Verschiebung. Folgendes Programm erlaubt jedoch eine schnelle Vertikalbewegung in BASIC, in der die Player auch um mehrere Zeilen nach oben oder unten verschoben werden können (Atari XL/XE/400/800).

Die Playfields und Player bestehen dabei aus Strings. Das Playfield ist 255 Byte lang (beziehungsweise 128 Byte bei zweizeiliger Auflösung), und die Lösung des Strings für die Player ist identisch mit der Anzahl der Zeilen des jeweiligen Players. Vor der Eingabe des Programmes sollten Sie den Computer aus- und wieder einschalten, damit auf jeden Fall alle Variablen aus der Variablen-Wert-Tabelle dem Variablen-Feld gelöscht sind.

In den Zeilen 10–30 werden alle nötigen Variablen dimensioniert. Es sollen vier Player aktiviert werden, die jeweils elf Zeilen hoch, also elf Byte in den Strings belegen (VO1\$–VO4\$). Außerdem werden vier 255 Byte lange Strings für die Playfields benötigt (PF1\$–PF4\$). Der String LOESCH\$ überschreibt die alten Daten in den Playfields. In den Zeilen 40–60 werden die Playerdaten anhand von einer ATASCII-Zeichenkette in die „Playerstrings“ eingelesen. Das oberste Ende des freien Speichers minus 1024 Byte, die die vier Playfields benötigen, wird in Zeile 80 ermittelt und anschließend in die Adresse 54279 geschrieben, welche die Anfangsadresse der Playfields benötigt. In Zeile 100 wird die Anfangsadresse der Variablen-Wert-Tabelle festgelegt. Diese Tabelle enthält alle Informationen über jede Variable, also zum Beispiel die Länge eines Strings, die legale Länge eines Strings (wird durch DIM festgelegt)

oder die Differenz zwischen der Adresse, wo die Daten eines Strings abgelegt sind und dem Beginn der Speicheradresse für Text und Felder (wird in Zeile 110 ermittelt). Letztere Information wird in den Zeilen 120 und 130 ermittelt und dann in die Variablen-Wert-Tabelle geschrieben (Zeile 140–200). Diese Änderung ist nötig, weil die vier Playerstrings am Ende des freien Speichers liegen und nicht im dafür vorgesehenen Speicherbereich für Text und Felder. Um noch einmal auf die Variablen-Wert-Tabelle zurückzukommen: Die Information einer Variablen ist immer acht Byte lang. Da die Variablen-Wert-Tabelle eine Größe von nur 1 KByte hat, kann das Atari-BASIC nicht mehr als 128 Variablen gleichzeitig verwalten.

Schließlich wird in Zeile 220 die Playergrafik für einfache Auflösung (POKE 559,46 für zweizeilige Auflösung) aktiviert. In den Zeilen 230–430 ist eine kleine Demonstration angehängt, die die schnelle Vertikalbewegung verdeutlicht.

Zusätzlich ist noch zu beachten, daß ober- und unterhalb des Players mehrere Leerzeilen stehenbleiben, damit bei einer Vertikalbewegung der Wert der letzten Zeile gelöscht wird und der Player keine „Fäden“ zieht. Zu der schnellen Vertikalbewegung kommt der Vorteil, daß Playerdaten blitzschnell ausgetauscht werden können (siehe Flügelbewegung).
Axel Küpper

500 Mark winken dem Gewinner

```

FC 10 DIM PF1$(255),PF2$(255),PF3$(255),PF4$(255)
OV 20 DIM VO1$(11),VO2$(11),VO3$(11),VO4$(11)
YC 30 DIM LOESCH$(255)
QE 35 GRAPHICS 17
SN 40 VO1$="*****L*****"
SB 50 VO2$="*****L f*****"
QT 60 VO3$=VO1$:VO4$=VO2$
SC 70 FOR Q=1 TO 255:LOESCH$(Q,Q)=CHR$(Q):NEXT Q
IH 80 MEMTOP=8*(INT(PEEK(742)/8)-0.5)
OO 90 POKE 54279,MEMTOP
RN 100 VUTP=PEEK(134)+PEEK(135)*256
IN 110 STARP=PEEK(140)+PEEK(141)*256
IY 120 PM=256*MEMTOP
JK 130 DISP=PM-STARP
IY 140 ADD=2
QB 150 FOR Q=1 TO 4
UO 160 PMH=INT(DISP/256):PML=DISP-256*PMH
JL 170 POKE VUTP+ADD,PML
ON 180 POKE VUTP+ADD+1,PMH
OZ 190 ADD=ADD+8:DISP=DISP+256
IU 200 NEXT Q
KZ 210 PF1$=LOESCH$:PF2$=LOESCH$:PF3$=LOESCH$:PF4$=LOESCH$
QL 220 POKE 559,62:POKE 53277,3
DF 230 X1=50:Y1=50
NR 240 X2=100:Y2=200
LO 250 X3=150:Y3=50
OP 260 X4=200:Y4=200
PF 270 POKE 53248,X1:PF1$(Y1)=VO1$
TK 280 POKE 53249,X2:PF2$(Y2)=VO2$
SS 290 POKE 53250,X3:PF3$(Y3)=VO3$
WE 300 POKE 53251,X4:PF4$(Y4)=VO4$
XB 310 POKE 704,12:POKE 705,27:POKE 706,42:POKE 707,57
EE 320 Y11=1:Y12=-1
EA 330 FOR Q=1 TO 150
YZ 340 Y1=Y1+Y11:Y2=Y2+Y12
CX 350 Y3=Y3+Y11:Y4=Y4+Y12
YF 360 PF1$(Y1)=VO1$
ZP 370 PF2$(Y2)=VO2$
AZ 380 PF3$(Y3)=VO3$
CJ 390 PF4$(Y4)=VO4$
YY 400 VO1$=VO2$:VO2$=VO3$:VO3$=VO4$:VO4$=VO2$
IY 410 NEXT Q
OZ 420 Y11=Y11*-1:Y12=Y12*-1
NR 430 GOTO 330
  
```